

Mode d'Emploi : Outil de Masquage Automatique des Colonnes (Excel)

Njee HETTIARACHCHI – 20 avril 2026

Introduction

Ce guide détaille la mise en place et l'utilisation d'un outil interactif permettant de masquer instantanément les colonnes vides suite à l'application d'un filtre. Cet outil optimise la visibilité des données en éliminant le défilement horizontal inutile.

Phase 1 : Installation de la Macro

Cette étape intègre le script VBA (Macro) qui gère l'intelligence du masquage. Elle ne doit être effectuée qu'une seule fois par fichier.

1. Ouvrez votre classeur Excel.
2. Accédez à l'éditeur de code en appuyant sur **Alt + F11**.
3. Dans le menu supérieur, allez dans **Insertion > Module**.
4. Copiez et collez le code suivant dans la fenêtre qui vient de s'ouvrir :

```
Sub Masquer_Colonnes_Vides_Filtrees_Dynamique()  
    Dim ws As Worksheet  
    Dim lastRow As Long, lastCol As Long  
    Dim i As Long  
    Dim rngVisible As Range  
    Dim cell As Range  
    Dim colEstVide As Boolean  
  
    ' =====  
    ' CONFIGURATION : A adapter si besoin  
    Dim ligneEnTete As Integer  
    ligneEnTete = 2 ' <-- Indique ici la ligne ou se trouvent les filtres  
  
    Dim ligneDonnees As Integer  
    ligneDonnees = ligneEnTete + 1 ' Les donnees commencent juste en dessous  
  
    Dim colonnesFixes As Integer  
    colonnesFixes = 9 ' <-- Indique le nb de colonnes a partir duquel filtrer  
    ' =====  
  
    Set ws = ActiveSheet  
    Application.ScreenUpdating = False  
  
    ' 1. Reafficher toutes les colonnes avant de tester  
    ws.Cells.EntireColumn.Hidden = False  
  
    ' 2. Trouver la derniere ligne et colonne  
    lastRow = ws.Cells(ws.Rows.Count, "A").End(xlUp).Row  
  
    ' On cherche la derniere colonne sur la ligne d'en-tete  
    lastCol = ws.Cells(ligneEnTete, ws.Columns.Count).End(xlToLeft).Column  
  
    ' 3. Identifier les lignes actuellement VISIBLES  
    On Error Resume Next  
    ' On commence la recherche a partir de la vraie ligne de donnees  
    Set rngVisible = ws.Range(ws.Cells(ligneDonnees, 1), ws.Cells(lastRow, 1)).SpecialCells(xlCellTypeVisible)  
    On Error GoTo 0  
  
    ' Securite si aucune ligne n'est visible  
    If rngVisible Is Nothing Then  
        Application.ScreenUpdating = True  
        Exit Sub  
    End If  
  
    ' 4. Boucle sur les colonnes a partir de la i-eme (I)  
    For i = colonnesFixes To lastCol  
        colEstVide = True ' On suppose que la colonne est vide par default  
  
        ' On ne parcourt que les lignes visibles  
        For Each cell In rngVisible.Cells  
            ' On lit la valeur de la cellule (en tenant compte de la colonne i)  
            If Trim(ws.Cells(cell.Row, i).Value) <> "" Then  
                colEstVide = False ' On a trouve une donnee ! Inutile de chercher plus loin  
                Exit For  
            End If  
        Next cell  
  
        ' Si la colonne est toujours "vide", on la masque  
        If colEstVide Then  
            ws.Columns(i).Hidden = True  
        End If  
    Next i  
  
    Application.ScreenUpdating = True  
End Sub
```

Phase 2 : Configuration du Raccourci Clavier

Afin de rendre l'utilisation fluide et d'éviter que des boutons ne disparaissent lors des exports, nous allons attribuer un raccourci clavier.

1. Fermez l'éditeur VBA et retournez sur votre feuille Excel.
2. Allez dans l'onglet **Affichage** (ou **Développeur**), puis cliquez sur **Macros**.
3. Sélectionnez la macro **Masquer_Colonne_Vides_Filtrees** dans la liste.
4. Cliquez sur le bouton **Options...**
5. Dans la case **Touche de raccourci**, maintenez la touche **Maj** enfoncée et tapez **M**. Le raccourci affiché doit devenir **Ctrl+Maj+M**.
6. Cliquez sur **OK**, puis fermez la fenêtre des macros.

Phase 3 : Utilisation au Quotidien

1. Appliquez vos filtres habituels sur les colonnes souhaitées (ex : Sous-famille, Type de pièce).
2. Appuyez simplement sur les touches **Ctrl+Maj+M** de votre clavier.
3. Le tableau se réduit instantanément aux colonnes contenant des données pertinentes.

Phase 4 : Personnalisation du Filtre

Pour personnaliser le comportement de la macro et modifier l'amplitude des colonnes fixes à ignorer :

1. Réouvrez votre fichier et appuyez sur **Alt + F11**.
2. Accédez à l'onglet **Module**.
3. Localisez la section **Configuration**
4. Modifiez la valeur du nombre de ligne ou de colonnes.
5. Faites un **CTRL + S** et fermez la page
6. Consultez si besoin la Macro dans **Affichage -> Afficher les Macros** pour s'assurer que le raccourci fonctionne toujours

Avertissement Important : Sauvegarde

1. Le fichier doit impérativement être enregistré au format **Classeur Excel prenant en charge les macros (*.xlsm)**. L'utilisation du format standard **.xlsx** entraînera la perte du code VBA.
2. Si l'étape du copier-coller fait apparaître des lignes en rouge, cela est certainement dû à des espaces qui se sont rajoutés entre des termes séparés par des points (ex : **Cell.Row** devenu **Cell . Row** au moment du recopiage), il suffit de les retirer et re-exécuter la Macro depuis le Menu **Affichage -> Afficher les Macros**.